<?xml version="1.0" encoding="UTF-8"?>
<xsl:stylesheet xmlns:xsl="http://www.w3.org/1999/XSL/Transform" version="1.0" xmlns="http://www.w3.org/1999/xhtml">
	<xsl:output encoding="UTF-8" method="html" indent="yes"/>
	<xsl:include href="common.xsl"/>
	<xsl:template match="/">
		<xsl:apply-templates/>
	</xsl:template>
	<xsl:template match="sqlzen">
		<xsl:variable name="title">
			<xsl:choose>
				<xsl:when test="$lang = 'ko'">'<xsl:value-of select="@targetName"/>' - SQLzen 도큐먼트</xsl:when>
				<xsl:otherwise>'<xsl:value-of select="@targetName"/>' - SQLzen document</xsl:otherwise>
			</xsl:choose>
		</xsl:variable>
		<html>
			<head>
				<title>
					<xsl:value-of select="$title"/>
				</title>
			</head>
			<body>
				<h1 align="center">
					<xsl:value-of select="$title"/>
				</h1>
				<xsl:apply-templates/>
			</body>
		</html>
	</xsl:template>
	<xsl:template match="rule">
		<h1>
			<xsl:choose>
				<xsl:when test="$lang = 'ko'">DB규칙</xsl:when>
				<xsl:otherwise>DB Rules</xsl:otherwise>
			</xsl:choose>
			<xsl:call-template name="putXmlIdText"/>
		</h1>
		<xsl:apply-templates/>
	</xsl:template>
	<xsl:template match="dictionary">
		<h1>
			<xsl:choose>
				<xsl:when test="$lang = 'ko'">용어집</xsl:when>
				<xsl:otherwise>Dictionary</xsl:otherwise>
			</xsl:choose>
			<xsl:call-template name="putXmlIdText"/>
		</h1>
		<xsl:if test="word">
			<h2>
				<xsl:choose>
					<xsl:when test="$lang ='ko'">단어</xsl:when>
					<xsl:otherwise>Words</xsl:otherwise>
				</xsl:choose>
			</h2>
			<table border="1">
				<tr>
					<th>
						<xsl:choose>
							<xsl:when test="$lang='ko'">단어</xsl:when>
							<xsl:otherwise>Word</xsl:otherwise>
						</xsl:choose>
					</th>
					<th>
						<xsl:choose>
							<xsl:when test="$lang='ko'">설명</xsl:when>
							<xsl:otherwise>Description</xsl:otherwise>
						</xsl:choose>
					</th>
				</tr>
				<xsl:apply-templates select="word"/>
			</table>
		</xsl:if>
		<xsl:if test="tableName">
			<h2>
				<xsl:choose>
					<xsl:when test="$lang ='ko'">테이블명</xsl:when>
					<xsl:otherwise>Table Names</xsl:otherwise>
				</xsl:choose>
			</h2>
			<table border="1">
				<tr>
					<th>
						<xsl:choose>
							<xsl:when test="$lang='ko'">테이블명</xsl:when>
							<xsl:otherwise>Table Name</xsl:otherwise>
						</xsl:choose>
					</th>
					<th>
						<xsl:choose>
							<xsl:when test="$lang='ko'">설명</xsl:when>
							<xsl:otherwise>Description</xsl:otherwise>
						</xsl:choose>
					</th>
				</tr>
				<xsl:apply-templates select="tableName"/>
			</table>
		</xsl:if>
		<xsl:if test="columnName">
			<h2>
				<xsl:choose>
					<xsl:when test="$lang ='ko'">컬럼명</xsl:when>
					<xsl:otherwise>Column Names</xsl:otherwise>
				</xsl:choose>
			</h2>
			<table border="1">
				<tr>
					<th>
						<xsl:choose>
							<xsl:when test="$lang='ko'">컬럼명</xsl:when>
							<xsl:otherwise>Column Name</xsl:otherwise>
						</xsl:choose>
					</th>
					<th>
						<xsl:choose>
							<xsl:when test="$lang='ko'">설명</xsl:when>
							<xsl:otherwise>Description</xsl:otherwise>
						</xsl:choose>
					</th>
				</tr>
				<xsl:apply-templates select="columnName"/>
			</table>
		</xsl:if>
	</xsl:template>
	<xsl:template match="word">
		<tr>
			<td>
				<xsl:value-of select="@name"/>
			</td>
			<td>
				<xsl:for-each select="comment[ $lang = @lang ]">
					<xsl:call-template name="putCommentText"/>
				</xsl:for-each>
			</td>
		</tr>
	</xsl:template>
	<xsl:template match="tableName | columnName">
		<tr>
			<td>
				<xsl:call-template name="putDbObjectName"/>
			</td>
			<td>
				<xsl:for-each select="comment[ $lang = @lang ]">
					<xsl:call-template name="putCommentText"/>
				</xsl:for-each>
			</td>
		</tr>
	</xsl:template>
	<xsl:template name="putXmlIdText">
		<xsl:if test="@xml:id">
			<xsl:text>( </xsl:text>
			<xsl:value-of select="@xml:id"/>
			<xsl:value-of select="@xmlns"/>
			<xsl:text> )</xsl:text>
		</xsl:if>
	</xsl:template>
	<xsl:template match="affixes">
		<h2>
			<xsl:choose>
				<xsl:when test="$lang = 'ko'">접사</xsl:when>
				<xsl:otherwise>Affixes</xsl:otherwise>
			</xsl:choose>
		</h2>
		<xsl:if test="entityPrefix">
			<xsl:call-template name="putEntityPrefix"/>
		</xsl:if>
		<xsl:if test="entitySuffix">
			<xsl:call-template name="putEntitySuffix"/>
		</xsl:if>
		<xsl:if test="attributePrefix">
			<xsl:call-template name="putAttributePrefix"/>
		</xsl:if>
		<xsl:if test="attributeSuffix">
			<xsl:call-template name="putAttributeSuffix"/>
		</xsl:if>
	</xsl:template>
	<xsl:template name="putEntitySuffix">
		<h3>
			<xsl:choose>
				<xsl:when test="$lang='ko'">엔터티 접미사</xsl:when>
				<xsl:otherwise>Entity Suffixes</xsl:otherwise>
			</xsl:choose>
		</h3>
		<table border="1">
			<tr>
				<th>
					<xsl:choose>
						<xsl:when test="$lang='ko'">접미사명</xsl:when>
						<xsl:otherwise>Suffix Name</xsl:otherwise>
					</xsl:choose>
				</th>
				<xsl:call-template name="putAffixTableHeader"/>
			</tr>
			<xsl:apply-templates select="entitySuffix"/>
		</table>
	</xsl:template>
	<xsl:template name="putEntityPrefix">
		<h3>
			<xsl:choose>
				<xsl:when test="$lang='ko'">엔터티 접두사</xsl:when>
				<xsl:otherwise>Entity Prefixes</xsl:otherwise>
			</xsl:choose>
		</h3>
		<table border="1">
			<tr>
				<th>
					<xsl:choose>
						<xsl:when test="$lang='ko'">접두사명</xsl:when>
						<xsl:otherwise>Prefix Name</xsl:otherwise>
					</xsl:choose>
				</th>
				<xsl:call-template name="putAffixTableHeader"/>
			</tr>
			<xsl:apply-templates select="entityPrefix"/>
		</table>
	</xsl:template>
	<xsl:template name="putAttributeSuffix">
		<h3>
			<xsl:choose>
				<xsl:when test="$lang='ko'">어트리뷰트 접미사</xsl:when>
				<xsl:otherwise>Attribute Suffixes</xsl:otherwise>
			</xsl:choose>
		</h3>
		<table border="1">
			<tr>
				<th>
					<xsl:choose>
						<xsl:when test="$lang='ko'">접미사명</xsl:when>
						<xsl:otherwise>Suffix Name</xsl:otherwise>
					</xsl:choose>
				</th>
				<xsl:call-template name="putAffixTableHeader"/>
			</tr>
			<xsl:apply-templates select="attributeSuffix"/>
		</table>
	</xsl:template>
	<xsl:template name="putAttributePrefix">
		<h3>
			<xsl:choose>
				<xsl:when test="$lang='ko'">어트리뷰트 접두사</xsl:when>
				<xsl:otherwise>Attribute Prefixes</xsl:otherwise>
			</xsl:choose>
		</h3>
		<table border="1">
			<tr>
				<th>
					<xsl:choose>
						<xsl:when test="$lang='ko'">접두사명</xsl:when>
						<xsl:otherwise>Prefix Name</xsl:otherwise>
					</xsl:choose>
				</th>
				<xsl:call-template name="putAffixTableHeader"/>
			</tr>
			<xsl:apply-templates select="attributePrefix"/>
		</table>
	</xsl:template>
	<xsl:template name="putAffixTableHeader">
		<xsl:call-template name="putCommentTableHeader"/>
	</xsl:template>
	<xsl:template name="putCommentTableHeader">
		<th>
			<xsl:choose>
				<xsl:when test="$lang='ko'">의미</xsl:when>
				<xsl:otherwise>Brief Meaning</xsl:otherwise>
			</xsl:choose>
		</th>
		<th>
			<xsl:choose>
				<xsl:when test="$lang='ko'">설명</xsl:when>
				<xsl:otherwise>Description</xsl:otherwise>
			</xsl:choose>
		</th>
	</xsl:template>
	<xsl:template match="entityPrefix | entitySuffix | attributePrefix | attributeSuffix">
		<tr>
			<td>
				<xsl:if test="contains( name(), 'Suffix' )">~</xsl:if>
				<xsl:value-of select="@name"/>
				<xsl:if test="contains( name(), 'Prefix' )">~</xsl:if>
			</td>
			<xsl:for-each select="comment[ $lang = @lang ]">
				<xsl:call-template name="putCommentTableBody"/>
			</xsl:for-each>
		</tr>
	</xsl:template>
	<xsl:template name="putCommentTableBody">
		<td>
			<xsl:value-of select="@brief"/>
		</td>
		<td>
			<xsl:value-of select="."/>
		</td>
	</xsl:template>
	<xsl:template match="dbObjects">
		<h3>
			<xsl:choose>
				<xsl:when test="$lang='ko'">DB객체접사</xsl:when>
				<xsl:otherwise>DB Object Affixes</xsl:otherwise>
			</xsl:choose>
		</h3>
		<table border="1">
			<tr>
				<th>
					<xsl:choose>
						<xsl:when test="$lang='ko'">객체분류</xsl:when>
						<xsl:otherwise>Object Type</xsl:otherwise>
					</xsl:choose>
				</th>
				<th>
					<xsl:choose>
						<xsl:when test="$lang='ko'">접두사</xsl:when>
						<xsl:otherwise>Prefix</xsl:otherwise>
					</xsl:choose>
				</th>
				<th>
					<xsl:choose>
						<xsl:when test="$lang='ko'">접미사</xsl:when>
						<xsl:otherwise>Suffix</xsl:otherwise>
					</xsl:choose>
				</th>
			</tr>
			<xsl:for-each select="*">
				<xsl:variable name="n" select="name()"/>
				<tr>
					<td>
						<xsl:choose>
							<xsl:when test="$n = 'dbLink'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">DB링크</xsl:when>
									<xsl:otherwise>DB Link</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'trigger'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">트리거</xsl:when>
									<xsl:otherwise>Trigger</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'procedure'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">프로시저</xsl:when>
									<xsl:otherwise>Procedure</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'function'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">함수</xsl:when>
									<xsl:otherwise>Function</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'package'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">패키지</xsl:when>
									<xsl:otherwise>Package</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'table'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">테이블</xsl:when>
									<xsl:otherwise>Table</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'view'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">뷰</xsl:when>
									<xsl:otherwise>View</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'primaryKey'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">기본키</xsl:when>
									<xsl:otherwise>Primary Key</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'foreignKey'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">외래키</xsl:when>
									<xsl:otherwise>Foreign Key</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'uniqueIndex'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">유니크 인덱스</xsl:when>
									<xsl:otherwise>Unique Index</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'normalIndex'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">노말 인덱스</xsl:when>
									<xsl:otherwise>Normal Index</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
							<xsl:when test="$n = 'sequence'">
								<xsl:choose>
									<xsl:when test="$lang = 'ko'">시퀀스</xsl:when>
									<xsl:otherwise>Sequence</xsl:otherwise>
								</xsl:choose>
							</xsl:when>
						</xsl:choose>
					</td>
					<td>
						<xsl:if test="@prefix">
							<xsl:value-of select="@prefix"/>
						</xsl:if>
					</td>
					<td>
						<xsl:if test="@suffix">
							<xsl:value-of select="@suffix"/>
						</xsl:if>
					</td>
				</tr>
			</xsl:for-each>
		</table>
	</xsl:template>
	<xsl:template match="tags">
		<h2>
			<xsl:choose>
				<xsl:when test="$lang = 'ko'">태그</xsl:when>
				<xsl:otherwise>Tags</xsl:otherwise>
			</xsl:choose>
		</h2>
		<xsl:if test="dbTag">
			<h3>
				<xsl:choose>
					<xsl:when test="$lang='ko'">DB태그</xsl:when>
					<xsl:otherwise>DB Tag</xsl:otherwise>
				</xsl:choose>
			</h3>
			<xsl:apply-templates select="dbTag"/>
		</xsl:if>
	</xsl:template>
	<xsl:template match="dbTag">
		<table border="1">
			<tr>
				<td>
					<xsl:choose>
						<xsl:when test="$lang='ko'">태그명</xsl:when>
						<xsl:otherwise>Tag Values</xsl:otherwise>
					</xsl:choose>
				</td>
				<td>
					<xsl:value-of select="@name"/>
					<xsl:text>( </xsl:text>
					<xsl:for-each select="comment[ @lang = $lang ]">
						<xsl:call-template name="putCommentText"/>
					</xsl:for-each>
					<xsl:text> )</xsl:text>
				</td>
			</tr>
			<tr>
				<td>
					<xsl:choose>
						<xsl:when test="$lang='ko'">태그값</xsl:when>
						<xsl:otherwise>Tag Values</xsl:otherwise>
					</xsl:choose>
				</td>
				<td>
					<xsl:for-each select="value">
						<xsl:value-of select="@name"/>
						<xsl:text>( </xsl:text>
						<xsl:for-each select="comment[$lang=@lang]">
							<xsl:call-template name="putCommentText"/>
						</xsl:for-each>
						<xsl:text> )</xsl:text>
						<xsl:if test="not( position() = last() )">, </xsl:if>
					</xsl:for-each>
				</td>
			</tr>
		</table>
	</xsl:template>
	<xsl:template match="forbiddenWords">
		<h2>
			<xsl:choose>
				<xsl:when test="$lang = 'ko'">금칙어</xsl:when>
				<xsl:otherwise>Forbidden Words</xsl:otherwise>
			</xsl:choose>
			<table border="1">
				<tr>
					<th>
						<xsl:choose>
							<xsl:when test="$lang='ko'">단어</xsl:when>
							<xsl:otherwise>Word</xsl:otherwise>
						</xsl:choose>
					</th>
					<th>
						<xsl:choose>
							<xsl:when test="$lang='ko'">설명</xsl:when>
							<xsl:otherwise>Description</xsl:otherwise>
						</xsl:choose>
					</th>
				</tr>
				<xsl:for-each select="word">
					<tr>
						<td>
							<xsl:value-of select="@name"/>
						</td>
						<xsl:for-each select="comment[ @lang = $lang ]">
							<td>
								<xsl:call-template name="putCommentText"/>
							</td>
						</xsl:for-each>
					</tr>
				</xsl:for-each>
			</table>
		</h2>
	</xsl:template>
	<xsl:template match="database">
		<h1>
			<xsl:choose>
				<xsl:when test="$lang = 'ko'">데이타베이스 정의</xsl:when>
				<xsl:otherwise>Database Definition</xsl:otherwise>
			</xsl:choose>
		</h1>
		<h3>
			<xsl:choose>
				<xsl:when test="$lang = 'ko'">정보태그</xsl:when>
				<xsl:otherwise>Infomation Tag</xsl:otherwise>
			</xsl:choose>
			<xsl:text> ( </xsl:text>
			<xsl:if test="@*[ contains( name(), 'Tag' ) ]">
				<xsl:for-each select="@*[ contains( name(), 'Tag' ) ]">
					<xsl:variable name="tn" select="substring-before( name(), 'Tag' )"/>
					<xsl:variable name="tv" select="."/>
					<xsl:for-each select="/sqlzen/rule/tags/dbTag[ @name = $tn ]">
						<xsl:for-each select="value">
							<xsl:if test="@name = $tv">
								<xsl:for-each select="comment[ $lang = @lang ]">
									<xsl:call-template name="putCommentText"/>
								</xsl:for-each>
							</xsl:if>
						</xsl:for-each>
					</xsl:for-each>
					<xsl:if test="not( position() = last() )">, </xsl:if>
				</xsl:for-each>
			</xsl:if>
			<xsl:text> ) </xsl:text>
		</h3>
		<xsl:apply-templates/>
	</xsl:template>
	<xsl:template match="table">
		<h2>
			<xsl:value-of select="@name"/>			
			<xsl:if test="comment">
				<xsl:text> ( </xsl:text>
				<xsl:for-each select="comment[ $lang = @lang ]">
					<xsl:call-template name="putCommentText"/>
				</xsl:for-each>				
				<xsl:text> ) </xsl:text>
			</xsl:if>
		</h2>
		<xsl:apply-templates/>
	</xsl:template>
	<xsl:template match="table/comment"/>
	<xsl:template match="table/relationInfo"/>
	<xsl:template match="columns">
		<xsl:if test="column">
			<h3>
				<xsl:choose>
					<xsl:when test="$lang = 'ko'">컬럼</xsl:when>
					<xsl:otherwise>Columns</xsl:otherwise>
				</xsl:choose>				
			</h3>
			<table border="1">
				<tr>
					<th>
						<xsl:choose>
							<xsl:when test="$lang = 'ko'">이름</xsl:when>
							<xsl:otherwise>Name</xsl:otherwise>
						</xsl:choose>
					</th>
					<th>
						<xsl:choose>
							<xsl:when test="$lang = 'ko'">타입</xsl:when>
							<xsl:otherwise>Type</xsl:otherwise>
						</xsl:choose>
					</th>
					<th>
						<xsl:choose>
							<xsl:when test="$lang = 'ko'">NULL아님</xsl:when>
							<xsl:otherwise>Not Null</xsl:otherwise>
						</xsl:choose>
					</th>
					<th>
						<xsl:choose>
							<xsl:when test="$lang = 'ko'">자동증가</xsl:when>
							<xsl:otherwise>Auto Increment</xsl:otherwise>
						</xsl:choose>
					</th>
					<th>
						<xsl:choose>
							<xsl:when test="$lang = 'ko'">기본값</xsl:when>
							<xsl:otherwise>Default Value</xsl:otherwise>
						</xsl:choose>
					</th>
				</tr>
				<xsl:apply-templates/>
			</table>
		</xsl:if>
	</xsl:template>
	<xsl:template match="column">
		<tr>
			<td>
				<xsl:value-of select="@name"/>
			</td>
			<td>
				<xsl:call-template name="putColumnType"/>
			</td>
			<td>
				<xsl:choose>
					<xsl:when test="@notNull = 'true'">O</xsl:when>
					<xsl:otherwise>X</xsl:otherwise>
				</xsl:choose>
			</td>
			<td>
				<xsl:choose>
					<xsl:when test="@autoIncrement = 'true'">O</xsl:when>
					<xsl:otherwise>X</xsl:otherwise>
				</xsl:choose>
			</td>
			<td>
				<xsl:value-of select="@default"/>
			</td>
		</tr>
	</xsl:template>
	<xsl:template match="constraints">		
			<h3>
				<xsl:choose>
					<xsl:when test="$lang = 'ko'">제약조건</xsl:when>
					<xsl:otherwise>Constraints</xsl:otherwise>
				</xsl:choose>				
			</h3>
			<table border="1">
				<tr>
					<th>
						<xsl:choose>
							<xsl:when test="$lang = 'ko'">타입</xsl:when>
							<xsl:otherwise>Type</xsl:otherwise>
						</xsl:choose>
					</th>
					<th>
						<xsl:choose>
							<xsl:when test="$lang = 'ko'">대상컬럼</xsl:when>
							<xsl:otherwise>Target Column</xsl:otherwise>
						</xsl:choose>
					</th>					
					<th>
						<xsl:choose>
							<xsl:when test="$lang = 'ko'">이름</xsl:when>
							<xsl:otherwise>Name</xsl:otherwise>
						</xsl:choose>
					</th>
					<th>
						<xsl:choose>
							<xsl:when test="$lang = 'ko'">참조컬럼</xsl:when>
							<xsl:otherwise>Refer Column</xsl:otherwise>
						</xsl:choose>
					</th>
				</tr>
				<xsl:apply-templates/>
			</table>		
	</xsl:template>
	<xsl:template match="primaryKey">
		<tr>
			<td>
				<xsl:choose>
					<xsl:when test="$lang = 'ko'">기본키</xsl:when>
					<xsl:otherwise>Primary Key</xsl:otherwise>
				</xsl:choose>				
			</td>
			<td>
				<xsl:value-of select="substring-after( @column, '.' )"/>
			</td>
			<td>
				<xsl:call-template name="putPrimaryKeyName"/>
			</td>
		</tr>
	</xsl:template>
	<xsl:template match="foreignKey">
		<tr>
			<td>
				<xsl:choose>
					<xsl:when test="$lang = 'ko'">외래키</xsl:when>
					<xsl:otherwise>Foreign Key</xsl:otherwise>
				</xsl:choose>				
			</td>
			<td>
				<xsl:value-of select="substring-after( @column, '.' )"/>
			</td>
			<td>
				<xsl:call-template name="putForeignKeyName"/>
			</td>
			<td>
				<xsl:call-template name="putReferenceTableColumn">
					<xsl:with-param name="name" select="@referColumn"/>
				</xsl:call-template>				
			</td>
		</tr>
	</xsl:template>
	<xsl:template match="unique">
		<tr>
			<td>
				<xsl:choose>
					<xsl:when test="$lang = 'ko'">유니크</xsl:when>
					<xsl:otherwise>Unique</xsl:otherwise>
				</xsl:choose>				
			</td>
			<td>
				<xsl:call-template name="putUniqueIndexColumn"/>
			</td>
			<td>
				<xsl:call-template name="putUniqueIndexName"/>
			</td>
		</tr>
	</xsl:template>	
	<xsl:template match="index">
		<tr>
			<td>
				<xsl:choose>
					<xsl:when test="$lang = 'ko'">인덱스</xsl:when>
					<xsl:otherwise>Index</xsl:otherwise>
				</xsl:choose>				
			</td>
			<td>
				<xsl:call-template name="putNormalIndexColumn"/>
			</td>
			<td>
				<xsl:call-template name="putNormalIndexName"/>
			</td>
		</tr>
	</xsl:template>
	<xsl:template match="*">
		<xsl:text>&#10;</xsl:text>
		<xsl:text>Unmatched element found : '</xsl:text>
		<xsl:value-of select="name(..)"/>
		<xsl:text>/</xsl:text>
		<xsl:value-of select="name()"/>
		<xsl:text>'</xsl:text>
	</xsl:template>
	<xsl:template match="text()"/>
</xsl:stylesheet>
